Wronskin determinantin differentiaaliyhtälön johtaminen

Wronskin determinantti määritellään homogeenisen lineaariyhtälön ratkaisujen ja niiden derivaattojen muodostamana determinanttina, mutta sille voidaan johtaa yksinkertainen differentiaaliyhtälö, joka näyttää, miten determinantti riippuu vain yhdestä differentiaaliyhtälössä olevasta kerroinfunktiosta. Johto on seuraavassa toteutettuna Mathematican keinoin. Lukija miettiköön, millainen tehtävä olisi käsin ratkaistuna (kuten toisen kertaluvun osalta on differentiaaliyhtälöiden kursseissa perinteisesti tehty).

Syötetään aluksi kertaluku ja muodostetaan vastaava normaalimuotoinen homogeeninen differentiaaliyhtälö, jossa kerroinfunktioita merkitään P_0, P_1 jne. Differentiaaliyhtälössä korkeimman kertaluvun termi annetaan erikseen, muut voidaan antaa summalausekkeena.

```
n = 3
```

```
3
```

```
diffyht = D[y[x], {x, n}] + Sum[Pk[x] D[y[x], {x, k}], {k, 0, n - 1}] == 0
```

$$y[x]\,P_0[x] + P_1[x]\,y'[x] + P_2[x]\,y''[x] + y^{(3)}[x] == 0$$

Olkoot y_1, y_2, ..., y_n differentiaaliyhtälön ratkaisuja. Sijoitetaan nämä yhtälöön, jolloin saadaan seuraavat yhtälöt:

```
diffyhtsij = Table[diffyht /. y -> yk, {k, 1, n}]
```

$$\{P_0[x]\,y_1[x] + P_1[x]\,y_1'[x] + P_2[x]\,y_1''[x] + y_1^{(3)}[x] == 0,$$
$$P_0[x]\,y_2[x] + P_1[x]\,y_2'[x] + P_2[x]\,y_2''[x] + y_2^{(3)}[x] == 0,$$
$$P_0[x]\,y_3[x] + P_1[x]\,y_3'[x] + P_2[x]\,y_3''[x] + y_3^{(3)}[x] == 0\}$$

Nämä ovat voimassa kaikilla arvoilla x.

Kerätään ratkaisut listaksi ja tätä derivoimalla muodostetaan vastaavat derivaattojen muodostamat listat. Kun nämä kerätään matriisiksi, saadaan Wronskin determinanttia vastaava matriisi.

```
ratk = Table[yk[x], {k, 1, n}]
```

$$\{y_1[x],\ y_2[x],\ y_3[x]\}$$

```
wronskimatr = Table[D[ratk, {x, k}], {k, 0, n - 1}]
```

$$\{\{y_1[x],\ y_2[x],\ y_3[x]\},\ \{y_1'[x],\ y_2'[x],\ y_3'[x]\},\ \{y_1''[x],\ y_2''[x],\ y_3''[x]\}\}$$

Matriisi on Mathematicassa listojen lista, mutta se voidaan tulostaa myös havainnollisemmassa muodossa:

```
wronskimatr // MatrixForm
```

$$\begin{pmatrix} y_1[x] & y_2[x] & y_3[x] \\ y_1'[x] & y_2'[x] & y_3'[x] \\ y_1''[x] & y_2''[x] & y_3''[x] \end{pmatrix}$$

Wronskin determinantti on saadun matriisin determinantti. Tämä kehitetään ja sijoitetaan oikeaksi puoleksi yhtälöön, jonka vasempana puolena on Wronskin determinantin symboli (jolle differentiaaliyhtälöä haetaan):

```
wronskidet = w[x] == Det[wronskimatr]
```

$$w[x] == -y_3[x]\,y_2'[x]\,y_1''[x] + y_2[x]\,y_3'[x]\,y_1''[x] + y_3[x]\,y_1'[x]\,y_2''[x] - y_1[x]\,y_3'[x]\,y_2''[x] - y_2[x]\,y_1'[x]\,y_3''[x] + y_1[x]\,y_2'[x]\,y_3''[x]$$

Derivoimalla saadaan vastaava Wronskin determinantin derivaattaa koskeva yhtälö:

```
wronskidetder = D[wronskidet, x]
```

$$w'[x] == -y_3[x]\,y_2'[x]\,y_1^{(3)}[x] + y_2[x]\,y_3'[x]\,y_1^{(3)}[x] + y_3[x]\,y_1'[x]\,y_2^{(3)}[x] - y_1[x]\,y_3'[x]\,y_2^{(3)}[x] - y_2[x]\,y_1'[x]\,y_3^{(3)}[x] + y_1[x]\,y_2'[x]\,y_3^{(3)}[x]$$

Wronskin determinantti ja sen derivaatta on saatu lausutuiksi ratkaisujen ja niiden derivaattojen avulla. Toisaalta ratkaisut toteuttavat alkuperäisen differentiaaliyhtälön. Jos näistä yhtälöistä voidaan eliminoida ratkaisufunktiot derivaattoineen, saadaan ehto, joka sitoo Wronskin determinantin, sen derivaatan ja differentiaaliyhtälön kerroinfunktiot. Ennalta ei ole selvää, että tällainen yhteys on olemassa.

Ehtoyhtälöitä on $n+2$ kappaletta:

```
yhtalot = Flatten[{diffyhtsij, wronskidet, wronskidetder}]
```

$$\{P_0[x]\,y_1[x] + P_1[x]\,y_1'[x] + P_2[x]\,y_1''[x] + y_1^{(3)}[x] == 0,$$
$$P_0[x]\,y_2[x] + P_1[x]\,y_2'[x] + P_2[x]\,y_2''[x] + y_2^{(3)}[x] == 0,$$
$$P_0[x]\,y_3[x] + P_1[x]\,y_3'[x] + P_2[x]\,y_3''[x] + y_3^{(3)}[x] == 0,$$
$$w[x] == -y_3[x]\,y_2'[x]\,y_1''[x] + y_2[x]\,y_3'[x]\,y_1''[x] + y_3[x]\,y_1'[x]\,y_2''[x] - y_1[x]\,y_3'[x]\,y_2''[x] - y_2[x]\,y_1'[x]\,y_3''[x] + y_1[x]\,y_2'[x]\,y_3''[x],$$
$$w'[x] == -y_3[x]\,y_2'[x]\,y_1^{(3)}[x] + y_2[x]\,y_3'[x]\,y_1^{(3)}[x] + y_3[x]\,y_1'[x]\,y_2^{(3)}[x] - y_1[x]\,y_3'[x]\,y_2^{(3)}[x] - y_2[x]\,y_1'[x]\,y_3^{(3)}[x] + y_1[x]\,y_2'[x]\,y_3^{(3)}[x]\}$$

Eliminoitavia symboleja ovat ratkaisut ja näiden derivaatat; yhteensä $n(n+1)$ kappaletta:

```
eliminoitavat = Flatten[Table[D[ratk, {x, k}], {k, 0, n}]]
```

$$\{y_1[x], y_2[x], y_3[x], y_1'[x], y_2'[x], y_3'[x], y_1''[x], y_2''[x], y_3''[x], y_1^{(3)}[x], y_2^{(3)}[x], y_3^{(3)}[x]\}$$

Eliminointi onnistuu, ja sen tuloksena saadaan ensimmäisen kertaluvun differentiaaliyhtälö Wronskin determinantille:

```
wronskidiffyht = Eliminate[yhtalot, eliminoitavat]
```

$$-w'[x] == w[x]\,P_2[x]$$

Differentiaaliyhtälö voidaan ratkaista:

```
DSolve[wronskidiffyht, w[x], x]
```

$\left\{\left\{w[x] \to e^{\int_1^x -P_2[K\$131]\, dK\$131}\, C[1]\right\}\right\}$

Huomaa, että Wronskin determinantti riippuu vain toiseksi korkeinta kertalukua olevan derivaatan kerroinfunktiosta. Integroimismuuttuja on Mathematican sisäisesti käyttöön ottama, ja sillä on tämän johdosta käyttäjän muuttujista varmasti eroava nimi.

Lukija voi muuttaa alussa annettua differentiaaliyhtälön kertalukua ja tutkia, miten Wronskin determinantin differentiaaliyhtälön etsiminen tällöin sujuu.

Linkit

Wronskin determinantti
homogeeniyhtälön ratkaisujoukko

SKK 27.04.2001